

# Teaching Intuitionistic and Classical Propositional Logic Using Isabelle

Jørgen Villadsen      Asta Halkjær From

DTU Compute - Department of Applied Mathematics and Computer Science - Technical University of Denmark

Patrick Blackburn

Section of Philosophy and Science Studies, IKH, Roskilde University, Denmark

We describe a natural deduction formalization of intuitionistic and classical propositional logic in the Isabelle/Pure framework. In contrast to earlier work, where we explored the pedagogical benefits of using a deep embedding approach to logical modelling, here we employ a logical framework modelling. This gives rise to simple and natural teaching examples and we report on the role it played in teaching our Automated Reasoning course in 2020 and 2021.

## 1 Introduction

We describe a formalization of intuitionistic and classical propositional logic in the Isabelle/Pure framework [3, 13, 17]. In a few lines of code we axiomatize our choice of connectives with their natural deduction introduction and elimination rules. This gives meaning to the (object) logic via the (meta) Isabelle/Pure framework rather than in potentially ambiguous natural language. Unlike pen-and-paper proofs that must be manually checked for mistakes, Isabelle can check automatically, reducing the burden on lecturers and teaching assistants. As we will see, by decorating the rules with appropriate attributes, we never have to mention exactly which rule is being applied, as the context determines it. This results in proofs that are natural to write and natural to read.

Our work is based on the Pure/Examples by Makarius in the Isabelle sources:

<https://isabelle.in.tum.de/dist/library/Pure/Pure-Examples/document.pdf>

That document, however, contains few comments and we have polished the formalization and tested it in class. Our formalization is available online in various formats:

[https://hol.compute.dtu.dk/Pure\\_I/](https://hol.compute.dtu.dk/Pure_I/)

We have used our approach in our Automated Reasoning course in 2020 and 2021:

<https://kurser.dtu.dk/course/02256>

This is an advanced MSc course (about 40 students) with a focus on the natural deduction proof system, first-order logic, higher-order logic and type theory, in particular Isabelle/HOL [12].

We call the approach Pure-I (“pure first”) for two reasons:

- We use Isabelle/Pure as a stepping stone towards the default higher-order logic in Isabelle/HOL
- We can understand the letter “I” both as a reference to the use of Isabelle by the students (the tools are a series of Isabelle theory files) but also for Intuitionism since we let the students add the Law of Excluded Middle (LEM) as the axiom  $p \vee \neg p$  when needed for classical reasoning.

```

theory Exam_Problem_4 imports Exam_Problem_4_Import begin

text <
  Replace the "oops" with a proper proof for each of the following formulas.
>

section <Question 4.1>

proposition <((p  $\longrightarrow$  q)  $\longrightarrow$  p)  $\longrightarrow$  p>
oops

— <... the rest of the exam problem has been omitted ...>

end

```

Figure 1: Part of an exam question using our formalization.

We note that Isabelle/HOL is built on classical logic but with a built-in proof method for intuitionistic logic (*iprover*).

We focus on propositional logic but extensions to first-order logic as well as higher-order logic are possible. We use the same natural deduction rule names as in our natural deduction assistant NaDeA [15]. We teach NaDeA first but this is not a requirement. However, by teaching NaDeA first we can move quicker through the material as the rules are already familiar to the students and only the Isabelle setting is new. This would also be possible if the students had already learned natural deduction in some other way.

Ideally, our students participating in the MSc course have already taken our advanced BSc course on Logical Systems and Logic Programming (about 80 students):

<https://kurser.dtu.dk/course/02156>

In practice this is not the case, in particular due to a large intake of international students, so we start the MSc course with some weeks on axiomatic systems and/or sequent calculus presented in a different way than in the BSc course [9, 16].

Figure 1 shows an actual question for the written exam, 27 May 2021, in our course 02256 Automated Reasoning. The solution has since been added as theorem *Peirce* in Fig. 3.

The exam was 2 hours with all aids. 36 students registered and 30 students passed.

The distribution of the grades is available online:

<https://karakterer.dtu.dk/Histogram/1/02256/Summer-2021>

In Denmark we use a so-called 7-step-scale, designed to be compatible with the ECTS grading scale, with A=12, B=10, C=7, D=4, E=02, Fx=00 and F=-3. Two students were not graded because they did not show up at the exam. The grade average were between B and C. This is quite a high grade average but not unusual for advanced courses.

In the next section we provide an overview of our teaching and the present paper.

Table 1: Topics in our course along with relevant tasks (parentheses denote optional tasks).

Topic	Section	Figures	Tasks
Logical Frameworks	3		Study the Isabelle/Pure framework
Axiomatizations	4		Study the Pure-I formalization
Definitions	5		Study the derived rules of defined connectives
Examples	6	2	Study the Figure Do at least a dozen proofs in intuitionistic logic (Port to Isabelle/HOL: replace Pure-I with Main <sup>1</sup> )
Classical Logic	7	3, 4	Study the first Figures Do a few proofs in classical logic (5, 6, 7, 8) (Do the proofs in the remaining Figures)
Exam preparation	-	1	Do at least a handful of proofs.

## 2 Overview of Teaching and Paper

We explain the following approaches in the following Section 3:

- Deep Embedding
- Shallow Embedding
- Logical Framework

Our use of Pure-I in the course proceeds in the following way. We ask students to install Isabelle before the course starts:

<https://isabelle.in.tum.de/>

As preparation, they can download and open Pure-I (see above) or if Isabelle is not installed, browse the formalization here:

[https://hol.compute.dtu.dk/Pure\\_I/Pure\\_I.html](https://hol.compute.dtu.dk/Pure_I/Pure_I.html)

We then cover the topics in Table 1, which are described further in the listed sections and figures. We have also included tasks for the students that are relevant to understanding the material.

In Section 8 we take a step back and consider the pedagogical benefits of this approach from a more general perspective, before we conclude in Section 9.

---

<sup>1</sup>Main is the Isabelle theory that defines Isabelle/HOL whereas Pure-I is our axiomatization.

### 3 Logical Frameworks vs. Deep/Shallow Embeddings

In this paper we present a simple and direct way of teaching propositional logic, namely by “axiomatizing” the Isabelle/Pure system in the most direct way possible. The pedagogical benefits of this approach are summed up in the words ‘simple’ and ‘natural’. The object and metalevel deductions mirror each other clearly — and the way that teachers would typically instruct students to construct proofs is directly reflected in the code the students learn to write to prove things in Isabelle (cf. Section 6).

We have used this system as part of a course on automated reasoning taught at the Technical University of Denmark (DTU); the course aimed to teach both natural deduction and Isabelle. It is the latest in a line of systems (based on courses taught at DTU) which explores different ways in which Isabelle can be used for pedagogical purposes [7, 8, 14, 15]. For example, in the paper presented at the 2020 edition of ThEdu, we described a formalization of a simple axiomatization, called W [8], for classical propositional logic. This formalization was built using a *deep embedding* of System W in the higher-order logic of Isabelle/HOL:

**Deep Embedding**      The act of representing one language, typically a logic or programming language, with another by modeling expressions in the former as data in the latter.

– Wiktionary, [https://en.wiktionary.org/wiki/deep\\_embedding](https://en.wiktionary.org/wiki/deep_embedding)

Deep embedding is certainly more appropriate when it comes to teaching many aspects of metatheory — it forces the student to get to grips with what is involved in formalizing a language *and its semantics*, and working in this way opens a natural path to studying such topics as soundness, completeness and interpolation (and learning how to use Isabelle/HOL to prove such results). This has certain pedagogical advantages — we argued in the paper that this approach forces students to consider metatheoretic issues right from the start. Deep embeddings are good for this purpose: for example, they make it possible to teach structural induction on the syntax (which is a key metalogical proof tool) as the object language syntax has a concrete representation in the metalogic.

Here, however, we want to take a step back from deep embedding. We want a gentler approach better suited for a first course in logic, especially for students with less technical backgrounds. Part of the goal is to teach *logic* (rather than metalogic) and in particular to teach natural deduction. The simple logical framework approach used in this paper enables us to directly “cook up” our Isabelle/Pure formalization from scratch without worrying about higher-order logic but while still teaching skills that are transferable to Isabelle/HOL.

We see the chief benefits of this lighter approach as being simplicity and elegance. This simplicity and elegance traces to two sources: the fact that we are working with natural deduction and the ease with which such proof rules can be reflected in Isabelle/Pure.

We start by introducing intuitionistic propositional logic and only afterwards axiomatize the Law of Excluded Middle (LEM). Intuitionistic propositional logic is simple in a number of ways. For example, every statement of the form  $\varphi \vee \psi$  is effectively transformable into an intuitionistic proof of  $\varphi$  or an intuitionistic proof of  $\psi$ . Moreover, intuitionistic propositional logic is effectively decidable: a finite constructive process applies uniformly to every formula, and this process will either produce an intuitionistic proof of the formula or demonstrate that no such proof can exist. When working in the intuitionistic fragment, our axiomatization embodies this process. In other words: it directly reflects, in very simple code, the process of natural deduction proofs.

We then add some complexity by moving to classical propositional logic. Working with intuitionistic proofs has built familiarity with the system, preparing the student to handle the more complicated clas-

sical proofs where there is no longer a clear recipe for how to prove a formula. As we will see, however, the move to classical logic does not require any new techniques and can therefore easily be explained.

Besides the logical framework approach taken here and the deep embeddings considered previously [8], there remains a third option to explore:

**Shallow Embedding** The act of representing one logic or language with another by providing a syntactic translation.

– Wiktionary, [https://en.wiktionary.org/wiki/shallow\\_embedding](https://en.wiktionary.org/wiki/shallow_embedding)

Shallow embeddings are another approach to formalizing logics, one that is widely used in the Isabelle community. Moreover, it is an important way of thinking about many topics in logic in its own right. Modal logic, for example, can be thought of in terms of Kripke models (a deep embedding perspective) but some logicians prefer to emphasize the standard translation of modal logic into first-order logic (a shallow embedding perspective). It would be interesting to use a shallow embedding of modal logic in Isabelle/HOL as a teaching tool that emphasised the standard translation perspective on modal logic. In terms of modal logic and Isabelle/Pure, Dionísio et al. [4] have successfully used Isabelle/Pure to teach hybrid logic to students in an approach similar to ours. The students even define their own labelled modal logic [2] in the end.

To conclude, in both the deep and shallow embedding approaches, we are forced to grapple with the complexity of Isabelle/HOL as our metalogic. Here, we only rely on the simple logical framework of Isabelle/Pure. This leaves more room for our goal: teaching logic using Isabelle.

## 4 Axiomatizations - Pure-I

In our approach we start at the bottom by declaring the type of formulas in our object logic, namely boolean:

```
typedecl bool
```

We need to link this type with Isabelle’s type of propositions using a so-called “truth judgment.” This automatically lifts our formulas so we can use the machinery provided by the Isabelle/Pure framework. The declaration is simple:

```
judgment Trueprop :: ( bool  $\Rightarrow$  prop ) (  $\langle$  -  $\rangle$  )
```

We are now ready to axiomatize our first connective:

```
axiomatization Imp (infixr  $\langle$   $\longrightarrow$   $\rangle$  3)
where Imp-I [intro]: ( p  $\Longrightarrow$  q )  $\Longrightarrow$  p  $\longrightarrow$  q
and Imp-E [elim]: ( p  $\longrightarrow$  q )  $\Longrightarrow$  p  $\Longrightarrow$  q
```

The first line declares a name, *Imp*, for our connective and an infix symbol,  $\longrightarrow$ , with right associativity and precedence 3. The next two lines describe the meaning via an introduction and an elimination rule. Here we see the power of using the Isabelle/Pure framework. Instead of using annotations, boxes or similar to discharge assumptions, we simply defer to the meta-logic’s implication,  $\Longrightarrow$ , to explain our object-logic’s implication. That is, to prove that *p* implies *q*, assume *p* at the meta level and prove *q*, letting the proof assistant handle the details. The elimination rule, modus ponens, is symmetric and lifts the object-level implication into the meta-level, letting us prove *q* from *p* when we know *p*  $\longrightarrow$  *q*.

We consider the axiomatization of disjunction next:

```

axiomatization Dis (infix  $\langle \vee \rangle$  4)
where Dis-E [elim]:  $\langle p \vee q \implies (p \implies r) \implies (q \implies r) \implies r \rangle$ 
and Dis-I1 [intro]:  $\langle p \implies p \vee q \rangle$ 
and Dis-I2 [intro]:  $\langle q \implies p \vee q \rangle$ 

```

Again we declare a name, *Dis*, and an infix symbol,  $\vee$ . This time we have one elimination rule and two introduction rules. The elimination rule allows us to conclude  $r$  from  $p \vee q$  if we can prove  $r$  from both  $p$  and  $q$ . Each of  $p$ ,  $q$  and  $r$  are meta-variables that can be instantiated for any formula. The two introduction rules allow us to form a disjunction when we have proved either side of it. The attributes on the rules, *intro* and *elim*, will allow us to use them without explicit reference in the proofs later on.

The axiomatization of conjunction uses the same principles as we have already seen:

```

axiomatization Con (infix  $\langle \wedge \rangle$  5)
where Con-I [intro]:  $\langle p \implies q \implies p \wedge q \rangle$ 
and Con-E1 [elim]:  $\langle p \wedge q \implies p \rangle$ 
and Con-E2 [elim]:  $\langle p \wedge q \implies q \rangle$ 

```

Finally we axiomatize falsity,  $\perp$ , with just an elimination rule, the principle of explosion:

```

axiomatization Falsity ( $\langle \perp \rangle$ )
where Falsity-E [elim]:  $\langle \perp \implies p \rangle$ 

```

Experienced Isabelle users will see that we have used it in a particularly simple and direct way: we have turned Isabelle into an “instance” of intuitionistic logic by adding axioms.

## 5 Definitions - Pure-I

Besides axiomatizations we can provide simple definitions that hide an expression under a name. In this way, we can define the dual of  $\perp$ , a symbol,  $\top$ , that denotes an always true formula:

```

definition Truth ( $\langle \top \rangle$ ) where  $\langle \top \equiv \perp \longrightarrow \perp \rangle$ 

```

We can still give a meaning to this, as an introduction rule, so we do not need to consider the implementation details later. Instead of being part of an axiomatization, this introduction rule is stated simply as a theorem:

```

theorem Truth-I [intro]:  $\langle \top \rangle$ 
unfolding Truth-def ..

```

Several concepts are at play here. In the first line we state the name of the theorem, the attributes and the goal  $\top$ . Next we unfold the definition, transforming the goal into  $\perp \longrightarrow \perp$ . The two periods `..` apply a matching introduction rule for the connective, in this case *Imp-I*. That reduces the goal to  $\perp \implies \perp$ , which is discharged automatically.

We define negation in the usual way, as implication of a contradiction:

```

definition Neg ( $\langle \neg \cdot \rangle$  [6] 6) where  $\langle \neg p \equiv p \longrightarrow \perp \rangle$ 

```

This brings us to a more interesting theorem and proof: from a formula and its negation we can prove anything. We start by stating the theorem and unfolding the definition:

```

theorem Neg-E [elim]:  $\langle \neg p \implies p \implies q \rangle$ 
unfolding Neg-def
proof –
assume  $\langle p \longrightarrow \perp \rangle$  and  $\langle p \rangle$ 
then have  $\langle \perp \rangle$  ..
then show  $\langle q \rangle$  ..
qed

```

The **proof** command starts the proof and the following hyphen tells Isabelle not to apply any initial proof rules to transform the goal (further). The **assume** command on the next line is implication introduction at the meta-level: it introduces the two  $\implies$  connectives in the goal. Since we unfolded the definition of negation, the first assumption is  $p \longrightarrow \perp$ , while the second one is  $p$  as expected. The **then** command makes these two assumptions (or any other just stated fact) available to the following command, which in this case is **have**. This command is used to state local results and here we use it to prove  $\perp$  from the two assumptions. The two periods **..** automatically apply the *Imp-E* rule to  $p \longrightarrow \perp$  and  $p$  in the proof context. The **show** on the penultimate line states a result like **have** but also discharges a goal, in our case  $q$ . Here the two periods automatically apply the *Falsity-E* rule. The final **qed** matches the initial **proof** and marks the end of the proof. It only “succeeds” if there are no remaining goals.

We define bi-implication as implication in both directions:

```

definition Iff (infixr  $\langle \longleftrightarrow \rangle$  3) where  $\langle p \longleftrightarrow q \equiv (p \longrightarrow q) \wedge (q \longrightarrow p) \rangle$ 

```

Again, we want to prove introduction and elimination rules that allow us to work with the connective without unfolding its definition. We start with the introduction rule. Given that we can prove  $q$  from  $p$  and  $p$  from  $q$  we seek to prove  $p \longleftrightarrow q$ :

```

theorem Iff-I [intro]:  $\langle (p \implies q) \implies (q \implies p) \implies p \longleftrightarrow q \rangle$ 
unfolding Iff-def
proof –
assume  $\langle p \implies q \rangle$  and  $\langle q \implies p \rangle$ 
from  $\langle p \implies q \rangle$  have  $\langle p \longrightarrow q \rangle$  ..
from  $\langle q \implies p \rangle$  have  $\langle q \longrightarrow p \rangle$  ..
from  $\langle p \longrightarrow q \rangle$  and  $\langle q \longrightarrow p \rangle$  show  $\langle (p \longrightarrow q) \wedge (q \longrightarrow p) \rangle$  ..
qed

```

We have already seen the **proof** and **assume** commands, which are used in the same way here. Instead of using **then**, we pick out the assumptions one at a time with **from** in the next two lines, introducing the object-logic implication in each direction. The final line combines the two implications into their conjunction: the definition of bi-implication.

The proofs of the two elimination rules only use techniques we have already seen, but some other axiomatized rules, making them suitable as exercises:

```

theorem Iff-E1 [elim]:  $\langle p \longleftrightarrow q \implies p \implies q \rangle$ 
unfolding Iff-def
proof –
assume  $\langle (p \longrightarrow q) \wedge (q \longrightarrow p) \rangle$ 
then have  $\langle p \longrightarrow q \rangle$  ..
then show  $\langle p \implies q \rangle$  ..
qed

theorem Iff-E2 [elim]:  $\langle p \longleftrightarrow q \implies q \implies p \rangle$ 
unfolding Iff-def

```

```

theory Pure_I_Verbose imports Pure_I begin

text <A result from theory Pure_I but with explicit rule names>

proposition < $p \longrightarrow \neg \neg p$ >
proof (rule Imp_I)
  assume < $p$ >
  show < $\neg \neg p$ >
  proof (rule Neg_I)
    assume < $\neg p$ >
    from < $p$ > and < $p$ > show < $\perp$ > by (rule Neg_E)
  qed
qed

end

```

Figure 2: A simple teaching example with explicit rule names in Isabelle/Pure.

```

proof —
  assume < $(p \longrightarrow q) \wedge (q \longrightarrow p)$ >
  then have < $q \longrightarrow p$ > ..
  then show < $q \Longrightarrow p$ > ..
qed

```

## 6 Selected Teaching Examples

Let us examine some examples. The **proposition** command is an alternative name for **theorem** and we use it for student exercises and examples. There is no difference to the system, only to the human reader.

```

proposition < $p \longrightarrow \neg \neg p$ >
proof
  assume < $p$ >
  show < $\neg \neg p$ >
  proof
    assume < $\neg p$ >
    from < $\neg p$ > and < $p$ > show < $\perp$ > ..
  qed
qed

```

Note the way the Isabelle/Pure code matches the way we would explain to students how they should go about proving this in natural deduction. It essentially spells out a constructive process for building the proof tree required — with the formula being proved if and only if such a proof tree exists.

In Fig. 2 we see the same example but with explicit rule names like *Imp\_I* and *Neg\_I* (and syntax highlighting as seen by the students). It is always possible to write explicitly which rule justifies a proof step, either for clarity or to convince oneself that the implicitly applied rule is the expected one. In general, however, we prefer to leave such rules implicit, so the structure of the proof stands out clearly.

Here is a proof of modus tollens. Once again, note how Isabelle/Pure tracks the way a teacher would explain to a beginner how to construct the relevant proof tree:

```

proposition < $(p \longrightarrow q) \wedge \neg q \longrightarrow \neg p$ >
proof
  assume < $(p \longrightarrow q) \wedge \neg q$ >
  show < $\neg p$ >
proof

```



```

assume ⟨p⟩
from ⟨(p → q) ∧ ¬q⟩ have ⟨p → q⟩ ..
from ⟨p → q⟩ and ⟨p⟩ have ⟨q⟩ ..
from ⟨(p → q) ∧ ¬q⟩ have ⟨¬q⟩ ..
from ⟨¬q⟩ and ⟨q⟩ show ⟨⊥⟩ ..
qed
qed

```

Here is a more complex example:

```

proposition ⟨(p ↔ q) ↔ q ↔ p⟩
proof
  assume ⟨p ↔ q⟩
  show ⟨q ↔ p⟩
  proof
    from ⟨p ↔ q⟩ show ⟨q ⇒ p⟩ ..
  next
    from ⟨p ↔ q⟩ show ⟨p ⇒ q⟩ ..
  qed
next
  assume ⟨q ↔ p⟩
  show ⟨p ↔ q⟩
  proof
    from ⟨q ↔ p⟩ show ⟨p ⇒ q⟩ ..
  next
    from ⟨q ↔ p⟩ show ⟨q ⇒ p⟩ ..
  qed
qed

```

Here, we need to prove both directions of the middle bi-implication by assuming the left-hand side and showing the right and vice versa. Conceptually, the **next** command switches to proving the remaining direction. Technically, it resets the local context, so we cannot use the assumption from the proof in one direction in the proof of the other direction. With this command at hand, the proof almost writes itself.

The above example works verbatim in Isabelle/HOL. This makes the skills obtained by working with this logic directly transferable to the larger setting of higher-order logic. The syntax is the same and the same rules are available, making this a stepping stone for proving things in Isabelle/HOL.

## 7 Classical Logic

Imagine we try to prove a formula that does not have a proof in our logic, e.g. Peirce's law:

$$((p \longrightarrow q) \longrightarrow p) \longrightarrow p$$

We can take a stab at it, applying introduction rules for the implications and so on, but we will never succeed. Like with pen and paper, the system does not provide any indication that no proof exists, but unlike with pen and paper we cannot mistakenly think one does by writing down a flawed proof. That specific formula does hold in classical propositional logic and so we could extend the axiomatization with the Law of Excluded Middle (LEM)  $p \vee \neg p$  to work in that logic instead. Then we would be able to find a proof.

```

theory Classical imports Pure_I begin

axiomatization where LEM: <p ∨ ¬ p>

theorem classical: <(¬ p ⇒ p) ⇒ p>
proof -
  have <p ⇒ p> .
  assume <¬ p ⇒ p>
  with LEM and <p ⇒ p> show <p> ..
qed

theorem Clavius: <(¬ p → p) → p>
proof
  assume <¬ p → p>
  show <p>
  proof (rule classical)
    assume <¬ p>
    with <¬ p → p> show <p> ..
  qed
qed

theorem Peirce: <((p → q) → p) → p>
proof
  assume <(p → q) → p>
  show <p>
  proof (rule classical)
    assume <¬ p>
    have <p → q>
    proof
      assume <p>
      with <¬ p> show <q> ..
    qed
    with <(p → q) → p> show <p> ..
  qed
qed
end

```

Figure 3: Axiomatization of the Law of Excluded Middle (LEM) with subsequent proofs of three other principles in classical logic.

In Fig. 3 we do so by axiomatizing the Law of Excluded Middle (LEM) on top of Pure-I. This axiomatization of a simple formula is easy to explain to students and clearly states an important point in classical logic: for every proposition, either this proposition or its negation is true. We first derive a more useful principle *classical*, named after its counterpart in Isabelle/HOL. The single period command proves immediate facts like  $p \Rightarrow p$ . The **with** command makes one or more assumptions or previously established facts available as justification for the following statement. We use the theorem *classical* to prove Clavius's law and Peirce's law; we can no longer rely only on two periods as for intuitionistic proofs, but must explicitly invoke a classical principle in the proof.

In Fig. 4 we give a wordy proof of the *classical* theorem from Fig. 3. Here we use the full name *this* for the single period proof method and *standard* for the double period. We also state *LEM* directly and use **from** instead of **with** to be more explicit. It is often useful for students to work with variants of the key proofs.

In Fig. 5 we consider the special NaDeA rule Boole in the Isabelle/Pure setting. After proving it from *classical* (and transitively *LEM*), we prove a variant of double negation elimination using it. Formulas like this can be used either for demonstration or as student exercises.

In Fig. 6 we consider classical contradiction (*ccontr* in Isabelle/HOL): if from  $\neg p$  we can derive falsity, then we can conclude  $p$ . We have already proved this as Boole when we recall that we have

```

theory Classical_Verbose imports Classical begin

text <The theorem "classical" from theory Classical with wordy proof>

theorem <( $\neg p \implies p$ )  $\implies p$ >
proof -
  have < $p \vee \neg p$ > by (rule LEM)
  have < $p \implies p$ > by this
  assume < $\neg p \implies p$ >
  from < $p \vee \neg p$ > and < $p \implies p$ > and < $\neg p \implies p$ > show < $p$ > by standard
qed

end

```

Figure 4: A proof from Fig. 3 with explicit rules.

defined  $\neg p \equiv p \longrightarrow \perp$ . We then prove the same formula as in Fig. 5 but stated using negation.

In Fig. 7 we define  $\perp$  and  $\top$  in Isabelle/HOL as *False* and *True*, respectively. We then define the Boole rule from the pre-defined *ccontr* by unfolding negation similarly to previously. Then we can insert our proof from Fig. 5 unmodified. This demonstrates how the skills obtained from working with our formalization in Isabelle/Pure are transferable to the full Isabelle/HOL.

In Fig. 8 we consider classical contradiction (*ccontr*) already available in Isabelle/HOL by repeating the proof from Fig. 6 in this setting.

As exercises we offer a number of formulas to be proved in intuitionistic or classical logic.

## 8 The Role of Proof Assistants

We have already mentioned a number of features of our course which we feel are pedagogically useful. In this section we want to stand back, take a more general perspective on our goals, and speculate a little about the future role of Isabelle/Pure (and other proof assistants) in the teaching of logic and mathematics.

Intuitionistic logic, perhaps more than any other logic, is interesting both philosophically and computationally. The key ideas of intuitionism are due to the Dutch mathematician Luitzen Egbertus Jan Brouwer, and they played a prominent role in the early 20th century debate on the foundations of mathematics. Brouwer was opposed to Russell’s logicism (Brouwer viewed logic as part of mathematics, not as a foundation for it) and also to Hilbert’s formalism (intuitionistic logic was formalized not by Brouwer himself but by later followers such as Heyting). Key themes driving Brouwer’s philosophy of mathematics were the activity of proof, and the construction of mathematical objects; these were conceived as mental activities of the “creative subject”, thus Brouwer attempted to side-step Platonism as well (the view that mathematics is the discovery of truths about some abstract, pre-existing metaphysical realm). As is well known, variants of Brouwer’s ideas have led to the developments of new logic(s) and have become a source of inspiration for later developments in constructive mathematics and program development. For more background on the technical, conceptual and historical underpinnings of intuitionism, see [1, 10, 11].

To state simply what we shall spell out in the following paragraphs: we believe that teaching intuitionistic logic early in a student’s career — and carefully drawing attention to the differences with

```

theory Classical_NaDeA imports Classical begin

theorem Boole: <(p  $\rightarrow$   $\perp$   $\implies$   $\perp$ )  $\implies$  p>
proof (rule classical)
  assume <p  $\rightarrow$   $\perp$   $\implies$   $\perp$ > and < $\neg$  p>
  have <p  $\rightarrow$   $\perp$ >
  proof
    assume <p>
    with < $\neg$  p> show < $\perp$ > ..
  qed
  with <p  $\rightarrow$   $\perp$   $\implies$   $\perp$ > have < $\perp$ > .
  then show <p> ..
qed

proposition <((p  $\rightarrow$   $\perp$ )  $\rightarrow$   $\perp$ )  $\rightarrow$  p>
proof
  assume <(p  $\rightarrow$   $\perp$ )  $\rightarrow$   $\perp$ >
  show <p>
  proof (rule Boole)
    assume <p  $\rightarrow$   $\perp$ >
    with <(p  $\rightarrow$   $\perp$ )  $\rightarrow$   $\perp$ > show < $\perp$ > ..
  qed
qed

end

```

Figure 5: A proof of the classical principle Boole used in NaDeA and a subsequent proof.

classical logic — is pedagogically useful because it sensitizes students to the nuances of proof. Moreover, we believe that doing so in the setting of Isabelle/Pure (or indeed, some other proof assistant) forces students to grapple more deeply with these distinctions, enables active experimentation, and prepares them for a future in which technology is likely to play an increasingly important role in logic and mathematics. Let us now try and explain why.

The constructive nature of intuitionistic proof is partially clarified in the BHK-interpretation (the Brouwer, Heyting, Kolmogorov interpretation) which contains clauses like the following:

A proof of  $\varphi \rightarrow \psi$  is any *construction* which *converts* any proof of  $\varphi$  into a proof of  $\psi$ .

Similarly:

A negated formula  $\neg\varphi$  is proved by giving a *construction* that *converts* a proof of  $\varphi$  into a proof of  $\perp$  (where  $\perp$  is unprovable).

The BHK-interpretation is an intuitive interpretation rather than a formal one, because it does not make clear what a “construction” (or “converts”) mean. But it is certainly suggestive. For example, any currently unproved mathematical statement, such as the Goldbach Conjecture (every even number greater than two is the sum of two primes) for which we have neither a proof, nor a proof of its negation, is a counterexample to LEM. Of course, we may lose some LEM counterexamples over time: Fermat’s Last Theorem used to be such a counterexample, but it stopped being one around September 1994, when Andrew Wiles had the crucial insight that lead him to the correct proof. That is, Brouwer’s view of mathematics emphasizes the dynamic role of creative construction, and the BHK-interpretation captures some of this. Furthermore, the words commonly used to describe the BHK-interpretation, like “construction”, “converts” or “transform” have a computational flavour. Indeed, one way of thinking about constructions is as programs: a proof of  $\varphi \rightarrow \psi$  is a program that takes a proof of  $\varphi$  as input and outputs a proof of  $\psi$ . This way of thinking is commonplace nowadays in various subfields of computer science, such as functional programming, and lies at the heart of what has become known as the Curry-Howard isomorphism.

```

theory Classical_NaDeA_Appendix imports Classical_NaDeA begin

text <The classical contradiction rule  $(\neg p \implies \perp) \implies p$  from Isabelle/HOL>

lemmas ccontr = Boole [folded Neg_def]

proposition < $\neg \neg p \implies p$ >
proof
  assume < $\neg \neg p$ >
  show < $p$ >
  proof (rule ccontr)
    assume < $\neg p$ >
    with < $\neg \neg p$ > show < $\perp$ > ..
  qed
qed
end

```

Figure 6: A classical proof using the principle Boole from NaDeA formulated as in Isabelle/HOL.

The late Michael Dummett, an influential English philosopher, published a well-known paper in 1975 (see [5], and also [6]) that attached philosophical significance to intuitionistic logic, but for reasons rather different from those emphasized by Brouwer and his followers. Brouwer emphasized the role of the “creative subject” – mathematics was some sort of non-linguistic mental activity. Dummett, on the other hand, strongly emphasizes the linguistic elements of mathematics; see [10] for a clear discussion. Dummett emphasizes that mathematics is a language that is to be *used* to be understood. Mathematics is about learning to *do* certain things – finding square roots, solving equations, differentiating functions, or proving theorems and corollaries. Learning the underlying logic of mathematics is all about learning what we need to do with logical statements — it’s about *process* — and the ability to recognize proofs and to appreciate their fine structure is a key part of this.

And this is the point where we feel the use of proof assistants such as Isabelle/Pure come in: *they open up new research and pedagogical spaces for this process*. Dummett’s paper is now almost half a century old, thus it precedes the era of interactive technologies which boost human cognitive capacities. Proof assistants are precisely such a technology: they open up new ways to explore the mathematical landscape via human-machine interaction. That is: what strikes us as interesting about Dummett’s work is the emphasis it places on linguistic/logical explorations underlying mathematics, and the activity of learning ways of proving. And proof assistants provide a truly novel way of zooming in on these activities.

Dummett may or may not be right to single out intuitionistic logic as special — we prefer to think in terms of *logics* of mathematics. But as a pedagogical tool, intuitionistic logic is undeniably useful. Teaching intuitionistic propositional logic early primes students to be aware of the nuances of mathematical proof. Doing so with the aid of a proof assistant, where they can actively compare classical and intuitionistic logics, amplifies the insights. It seems plausible that proof assistants will play an increasingly significant role in the development of mathematics and logic, turning research in new and unexpected directions. We should be training students in ways that will help them explore these new worlds, and our courses can be viewed as simple experiments in how to do this.

## 9 Concluding Remarks

We have shown how to use Isabelle/Pure as a logical framework for teaching intuitionistic and classical propositional logic. Instead of teaching each logic separately, we have shown first how to axiomatize an intuitionistic base and then how to add classical reasoning on top of it. Along these lines, we want

```

theory Main_HOL imports Main begin

text <The symbols  $\perp$  and  $\top$  are not part of the Isabelle/HOL theory Main>

notation False (< $\perp$ >) and True (< $\top$ >)

text <The Boole rule  $(p \rightarrow \perp \Rightarrow \perp) \Rightarrow p$  from theory Classical_NaDeA>

lemmas Boole = ccontr [unfolded not_def]

text <The example from theory Classical_NaDeA now works without changes>

proposition <(( $p \rightarrow \perp$ )  $\rightarrow \perp$ )  $\rightarrow p$ >
  — <The usual proof in Isabelle/HOL is just: by auto>
proof
  assume <( $p \rightarrow \perp$ )  $\rightarrow \perp$ >
  show < $p$ >
  proof (rule Boole)
    assume < $p \rightarrow \perp$ >
    with <( $p \rightarrow \perp$ )  $\rightarrow \perp$ > show < $\perp$ > ..
  qed
qed
end

```

Figure 7: A proof from Fig. 5 after a few definitions in Isabelle/HOL.

```

theory Main_HOL_Appendix imports Main_HOL begin

proposition < $\neg \neg p \rightarrow p$ >
  — <The usual proof in Isabelle/HOL is just: by auto>
proof
  assume < $\neg \neg p$ >
  show < $p$ >
  proof (rule ccontr)
    assume < $\neg p$ >
    with < $\neg \neg p$ > show < $\perp$ > ..
  qed
qed
end

```

Figure 8: A proof from Fig. 6 using the pre-defined *ccontr* in Isabelle/HOL.

students to learn a framework for describing natural deduction rules instead of only learning the rules themselves. We see this approach scaling from propositional logic to first-order logic and even further to higher-order logic. By defining each logic in the same logical framework we hope to make the distinctions between them clear to the students and want to explore this in future work.

## Acknowledgements

Thanks to Frederik Krogsdal Jacobsen and the anonymous reviewers for comments.

## References

- [1] Mark van Atten (2020): *Luitzen Egbertus Jan Brouwer*. In Edward N. Zalta, editor: *The Stanford Encyclopedia of Philosophy*, Spring 2020 edition, Metaphysics Research Lab, Stanford University.

- [2] David A. Basin, Seán Matthews & Luca Viganò (1998): *Labelled Modal Logics: Quantifiers*. *Journal of Logic, Language and Information* 7(3), pp. 237–263, doi:10.1023/A:1008278803780.
- [3] Stefan Berghofer & Makarius Wenzel (2008): *Logic-Free Reasoning in Isabelle/Isar*. In Serge Autexier, John A. Campbell, Julio Rubio, Volker Sorge, Masakazu Suzuki & Freek Wiedijk, editors: *Intelligent Computer Mathematics, 9th International Conference, AISC 2008, 15th Symposium, Calculemus 2008, 7th International Conference, MKM 2008, Birmingham, UK, July 28 - August 1, 2008. Proceedings, Lecture Notes in Computer Science* 5144, Springer, pp. 355–369, doi:10.1007/978-3-540-85110-3\_31.
- [4] F. Miguel Dionísio, Paula Gouveia & João Marcos (2005): *Defining and using deductive systems with Isabelle*. *Computing, Philosophy, and Cognition*, pp. 271–293.
- [5] Michael Dummett (1975): *The philosophical basis of intuitionistic logic*. In: *Studies in Logic and the Foundations of Mathematics*, 80, Elsevier, pp. 5–40.
- [6] Michael Dummett (2000): *Elements of intuitionism*. Oxford University Press.
- [7] Asta Halkjær From, Alexander Birch Jensen, Anders Schlichtkrull & Jørgen Villadsen (2019): *Teaching a Formalized Logical Calculus*. In Pedro Quaresma, Walther Neuper & João Marcos, editors: *Proceedings 8th International Workshop on Theorem Proving Components for Educational Software, ThEdu@CADE 2019, Natal, Brazil, 25th August 2019, EPTCS* 313, pp. 73–92, doi:10.4204/EPTCS.313.5.
- [8] Asta Halkjær From, Jørgen Villadsen & Patrick Blackburn (2020): *Isabelle/HOL as a Meta-Language for Teaching Logic*. In Pedro Quaresma, Walther Neuper & João Marcos, editors: *Proceedings 9th International Workshop on Theorem Proving Components for Educational Software, ThEdu@IJCAR 2020, Paris, France, 29th June 2020, EPTCS* 328, pp. 18–34, doi:10.4204/EPTCS.328.2.
- [9] Asta Halkjær From (2021): *Soundness and Completeness of an Axiomatic System for First-Order Logic*. *Archive of Formal Proofs*. [https://isa-afp.org/entries/FOL\\_Axiomatic.html](https://isa-afp.org/entries/FOL_Axiomatic.html), Formal proof development.
- [10] Rosalie Iemhoff (2020): *Intuitionism in the Philosophy of Mathematics*. In Edward N. Zalta, editor: *The Stanford Encyclopedia of Philosophy*, Fall 2020 edition, Metaphysics Research Lab, Stanford University.
- [11] Joan Moschovakis (2021): *Intuitionistic Logic*. In Edward N. Zalta, editor: *The Stanford Encyclopedia of Philosophy*, Fall 2021 edition, Metaphysics Research Lab, Stanford University.
- [12] Tobias Nipkow, Lawrence C. Paulson & Markus Wenzel (2002): *Isabelle/HOL — A Proof Assistant for Higher-Order Logic*. *Lecture Notes in Computer Science* 2283, Springer, doi:10.1007/3-540-45949-9.
- [13] Lawrence C. Paulson (1989): *The Foundation of a Generic Theorem Prover*. *J. Autom. Reason.* 5(3), pp. 363–397, doi:10.1007/BF00248324.
- [14] Anders Schlichtkrull, Jørgen Villadsen & Andreas Halkjær From (2018): *Students’ Proof Assistant (SPA)*. In Pedro Quaresma & Walther Neuper, editors: *Proceedings 7th International Workshop on Theorem proving components for Educational software, THedu@FLoC 2018, Oxford, United Kingdom, 18 July 2018, EPTCS* 290, pp. 1–13, doi:10.4204/EPTCS.290.1.
- [15] Jørgen Villadsen, Andreas Halkjær From & Anders Schlichtkrull (2017): *Natural Deduction and the Isabelle Proof Assistant*. In Pedro Quaresma & Walther Neuper, editors: *Proceedings 6th International Workshop on Theorem proving components for Educational software, ThEdu@CADE 2017, Gothenburg, Sweden, 6 Aug 2017, EPTCS* 267, pp. 140–155, doi:10.4204/EPTCS.267.9.
- [16] Jørgen Villadsen, Anders Schlichtkrull & Andreas Halkjær From (2018): *A Verified Simple Prover for First-Order Logic*. In Boris Konev, Josef Urban & Philipp Rümmer, editors: *Proceedings of the 6th Workshop on Practical Aspects of Automated Reasoning (PAAR 2018) co-located with Federated Logic Conference 2018 (FLoC 2018), Oxford, UK, 19 July 2018, CEUR Workshop Proceedings* 2162, CEUR-WS.org, pp. 88–104. Available at <http://ceur-ws.org/Vol-2162/paper-08.pdf>.
- [17] Makarius Wenzel, Lawrence C. Paulson & Tobias Nipkow (2008): *The Isabelle Framework*. In Otmane Aït Mohamed, César A. Muñoz & Sofiène Tahar, editors: *Theorem Proving in Higher Order Logics, 21st International Conference, TPHOLs 2008, Montreal, Canada, August 18-21, 2008. Proceedings, Lecture Notes in Computer Science* 5170, Springer, pp. 33–38, doi:10.1007/978-3-540-71067-7\_7.